Supplemental Materials

Measuring Physiological Influence in Dyads: A Guide to Designing, Implementing, and Analyzing Dyadic Physiological Studies

by K. Thorson et al., 2017, Psychological Methods

http://dx.doi.org/10.1037/met0000166
Graphing Values for Individual Dyad Members over Time

In the main text, we recommend graphing physiological values for individual dyad members over time to aid in the decision regarding interval length. If responding is relatively stable, shorter intervals may not be necessary. Here, we provide syntax for graphing individual estimates. In the first step, researchers need to sort their data by person (we use “id_1_00”). In the second step, researchers plot physiological values (here “pepreact_R”) by time and join the values by a straight line.

```sas
PROC SORT DATA=FMinfluence OUT=sort;
   BY id_1_00;
RUN;

SYMBOL1 COLOR=blue INTERPOL=join;
PROC GPLOT data=sort;
   PLOT pepreact_R*time / HMINOR = 0;
   BY id_1_00;
   run;
quit;
```

**Figure S1:** Syntax for graphing values for individual dyad members over time.

Lag Length

To create lagged receiver and sender physiology variables, the SAS syntax in Figure S2 can be used. Once the file is sorted by participant and time (line 3), lagged variables can be created (lines 4 and 6). Once the lagged variables are created, the variables should be marked as missing for the first time point (lines 5 and 7). Without this step, the lagged variables at the first time point would have the values of the previous participant at the last time point. For example, if Participant 3 had a PEP value of -5 at the last time point, Participant 4 would have a PEP value of -5 at the first time point. Marking the first time point as missing will eliminate this problem. In
lines 8 through 11 of this syntax, we look at the mean values for the lagged variables at every time point, making sure that there are no lagged data at the first time point.

If there are multiple phases within a study (e.g., a baseline interval, a high-conflict interaction, and then a low-conflict interaction), researchers should lag variables within each phase. To do this, the file should be sorted by participant and time, lagged variables should be created, and then values at the first time point of each new phase should be marked as missing. This ensures that values from one interval are not used to predict values within another interval.

```
DATA FMcontagion_3_notime10;               (1)
  SET data.FMcontagion_3_notime10;        (2)
  BY id_1_00 time;                        (3)
  pep_lag_R = lag(pepreact_R);            (4)
  IF first. id_1_00 THEN pep_lag_R = .;  (5)
  pep_lag_S = lag(pepreact_S);            (6)
  IF first. id_1_00 THEN pep_lag_S = .;  (7)
RUN;

PROC MEANS DATA=FMcontagion_3_notime10 noprint nway ;     (8)
CLASS time;                                                 (9)
VAR pep_lag_R pep_lag_S;                                    (10)
OUTPUT OUT=data2 MEAN=pep_lag_R_mean MEAN=pep_lag_S_mean;   (11)
RUN;
```

**Figure S2:** Syntax for lagging variables.

**Test of Distinguishability**

When a data set contains all distinguishable dyads (e.g., every dyad has one partner in an experimental condition and one partner in a control condition, or all dyads contain one female partner and one male partner), one can conduct a formal test of “distinguishability” for the fixed and random effects. In the basic model, there are three fixed effects that may differ as a function of the distinguishing factor: the intercept, the stability path, and the influence path. To test whether these parameters differ as a function of the distinguishing variable, one must include the main effect of the distinguishing variable and moderate the stability and influence paths by the distinguishing variable. In terms of random effects, partners may differ in the variances for the
intercept, the stability path, and the influence path all possible within-person and between-dyad members’ covariances may also differ for the types of partners. Lastly, there may also be “heterogeneity of error variances” for the two dyad members. We illustrate how to compare models in which these parameters are forced to be equal across partners, to those in which these parameters are allowed to vary as a function of the distinguishing variable, in the supplemental material. We encourage researchers to conduct this test to find the most parsimonious model. For example, a research may find that in a data set with heterosexual couples, the husbands and wives’ error variances are homogeneous (and forcing them to be equal does not worsen the fit of the model), and so estimating a pooled error variance simplifies the model.

To conduct a test of distinguishability, two models must be estimated: one model that treats dyads as distinguishable, and one that treats them as indistinguishable. Distinguishability can occur in the fixed effects and/or the random effects. Below we present the test of distinguishability using the data from Example 1 presented in the main text. In this example, gender is the distinguishing factor. We conduct a full test of distinguishability on the fixed and random effects at the same time (sometimes referred to as the omnibus test of distinguishability), but this test can be conducted to compare any effects in the model (e.g., just the fixed effects, just the random effects of the intercept, just the random effects of slopes, etc.).

In the two models presented in Figures S3 and S4, for the fixed effects, gender is included in three places: as a main effect, moderating the stability path, and moderating the influence path. These three effects are included in the example below in the syntax for the distinguishable case. Thus, there are three more fixed effects in the distinguishable case: gender, an interaction term of gender by the stability path, and an interaction term of gender by the influence path.
For the random effects, gender can be used to estimate separate variances and separate covariances. In the example below, we estimate the random effect of the intercept, the stability path, and the influence path. For illustration purposes, we have trimmed out the covariances, but these can easily be added back into the model (and the test of distinguishability can be done on these, too). The distinguishable model therefore estimates two intercept variances (one for women, one for men), two random effects of the stability path (one for women, one for men), two random effects of influence (one for women, one for men), and two error variances (using CSH—compound symmetry heterogeneous, which gives separate error variances for the two partners, as compared to CS—compound symmetry—in the indistinguishable model). There are therefore four more random effects estimated in the distinguishable case: an error variance, intercept variance, stability slope variance, and influence slope variance.

Taken together, two models will be estimated. The distinguishable model will test for moderation by the distinguishing factor at the level of the fixed effects and will also allow for separate error variances and covariances for men and women. In total, there are seven more parameters in the distinguishable model. Because these two models will be compared and changes have been made between them at the level of the fixed effects, ML (Maximum Likelihood) estimation must be used instead of the default method REML (Restricted Maximum Likelihood). The statement “METHOD = ML” can be added to the SAS syntax after “COVTEST.”

Once these models are estimated, a $\chi^2$ difference test can be computed by subtracting the deviances of the two models (called “ -2 Log Likelihood” under Fit Statistics). Because there is a difference of seven degrees of freedom between the distinguishable model and the indistinguishable model, a $\chi^2$ test with seven degrees of freedom can be estimated. One can then
look up the critical value of a $\chi^2$ with seven degrees of freedom (which is 14.067). The deviance for the distinguishable model is 23870.2, and the deviance for the indistinguishable model is 23925.1. The $\chi^2$ difference between these two models is 54.9, which, with 7 degrees of freedom, is statistically significant. Thus, we would need to treat dyads as distinguishable because setting parameter constraints that force dyads to be indistinguishable significantly worsened the fit of the model. If the $\chi^2$ difference test was not significant, dyads could be treated as indistinguishable as setting parameter constraints that force dyads to be indistinguishable would not have significantly worsened the fit of the model.

```
PROC MIXED COVTEST METHOD=ML;
CLASS dyad obs_id gender_class;
MODEL pepreact_R = gender pep_lag_RC pep_lag_SC
   gender*pep_lag_RC
   gender*pep_lag_SC
   /CL S DDFM=satterth;
RANDOM male female male*pep_lag_RC female*pep_lag_RC male*pep_lag_SC
   female*pep_lag_SC /SUB=dyad TYPE=VC;
REPEATED gender_class / TYPE=csh SUB=dyad*obs_id;
run;
```

**Figure S3:** Syntax for the distinguishable model.
Random Effects

In an ideal world, the random effects in one’s model would be fully saturated, resulting in 24 random effects for distinguishable dyads (six variances: parameters 1, 2, 3, 4, 5, and 6 in Table S1; six within-person covariances: 7, 8, 9, 10, 11, and 12; nine between-person covariances: 13, 14, 15, 16, 17, 18, 19, 20, and 21; two residual variances and one common covariance) or 14 random effects for indistinguishable dyads (three variances: parameters 1, 2, and 3 in Table S2; three within-person covariances: parameters 4, 5, and 6; six between-person covariances: 7, 8, 9, 10, 11, and 12; one residual variance and one common covariance).
**Table S1.** Possible random effects parameters in a stability and influence model with distinguishable dyads.

<table>
<thead>
<tr>
<th>Parameter</th>
<th>1.</th>
<th>2.</th>
<th>3.</th>
<th>4.</th>
<th>5.</th>
<th>6.</th>
</tr>
</thead>
<tbody>
<tr>
<td>1. Intercept for P1</td>
<td>1</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>2. Intercept for P2</td>
<td>13</td>
<td>2</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>3. Slope for receiver PEP (time_{t-1}) for P1</td>
<td>7</td>
<td>18</td>
<td>3</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>4. Slope for receiver PEP (time_{t-1}) for P2</td>
<td>16</td>
<td>10</td>
<td>14</td>
<td>4</td>
<td></td>
<td></td>
</tr>
<tr>
<td>5. Slope for sender PEP (time_{t-1}) for P1</td>
<td>8</td>
<td>19</td>
<td>9</td>
<td>21</td>
<td>5</td>
<td></td>
</tr>
<tr>
<td>6. Slope for sender PEP (time_{t-1}) for P2</td>
<td>17</td>
<td>11</td>
<td>20</td>
<td>12</td>
<td>15</td>
<td>6</td>
</tr>
</tbody>
</table>

**Table S2.** Possible random effects parameters in a stability and influence model with indistinguishable dyads.

<table>
<thead>
<tr>
<th>Parameter</th>
<th>1.</th>
<th>2.</th>
<th>3.</th>
<th>4.</th>
<th>5.</th>
<th>6.</th>
</tr>
</thead>
<tbody>
<tr>
<td>1. Intercept for P1</td>
<td>1</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>2. Intercept for P2</td>
<td>7</td>
<td>1</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>3. Slope for receiver PEP (time_{t-1}) for P1</td>
<td>4</td>
<td>10</td>
<td>2</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>4. Slope for receiver PEP (time_{t-1}) for P2</td>
<td>10</td>
<td>4</td>
<td>8</td>
<td>2</td>
<td></td>
<td></td>
</tr>
<tr>
<td>5. Slope for sender PEP (time_{t-1}) for P1</td>
<td>5</td>
<td>11</td>
<td>6</td>
<td>12</td>
<td>3</td>
<td></td>
</tr>
<tr>
<td>6. Slope for sender PEP (time_{t-1}) for P2</td>
<td>11</td>
<td>5</td>
<td>12</td>
<td>6</td>
<td>9</td>
<td>3</td>
</tr>
</tbody>
</table>

**Stability and Influence Model with Indistinguishable Dyads**

To demonstrate an analysis with indistinguishable dyads, we analyze data from a study similar to that of Example 1, where the only methodological difference is that all dyads were composed of two female participants. The primary difference in this analysis is that fixed and random effects are estimated to be the same across both members of the dyad (see Figures S5 and S6 for syntax). For the fixed effects, this is done by removing gender and any interactions with gender as predictor terms. For the random effects, this is done by creating a matrix which specifies constraints for the (co)variance parameters. In this example, we specify seven random
effects (see Table S3): variance in the intercepts (1), variance in receiver lag (2), between-person covariance of intercepts (3), between-person covariance of receiver lag (4), within-person covariance of intercepts with receiver lag (5), between-person covariance of intercepts with receiver lag (6), and variance of sender lag (7). Note that these random effects are the same as the ones specified in Example 1, but there are no longer separate random effects for each member of the dyad (as there were for men and women in Example 1).

Table S3. Variance covariance matrix with parameter constraints for the analysis of repeated measures indistinguishable dyadic data. P1 = partner 1; P2 = partner 2.

<table>
<thead>
<tr>
<th></th>
<th>1.</th>
<th>2.</th>
<th>3.</th>
<th>4.</th>
<th>5.</th>
<th>6.</th>
</tr>
</thead>
<tbody>
<tr>
<td>1. Intercept for P1</td>
<td>1</td>
<td>3</td>
<td>5</td>
<td>6</td>
<td></td>
<td></td>
</tr>
<tr>
<td>2. Intercept for P2</td>
<td></td>
<td>1</td>
<td>6</td>
<td>5</td>
<td></td>
<td></td>
</tr>
<tr>
<td>3. Slope for receiver PEP (time_{t-1}) for P1</td>
<td></td>
<td></td>
<td>2</td>
<td>4</td>
<td></td>
<td></td>
</tr>
<tr>
<td>4. Slope for receiver PEP (time_{t-1}) for P2</td>
<td></td>
<td></td>
<td></td>
<td>2</td>
<td></td>
<td></td>
</tr>
<tr>
<td>5. Slope for sender PEP (time_{t-1}) for P1</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td>7</td>
<td></td>
</tr>
<tr>
<td>6. Slope for sender PEP (time_{t-1}) for P2</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td>7</td>
</tr>
</tbody>
</table>

**Figure S5.** Syntax to set the parameter constraints outlined in Table S3.

data hhh;
input parm row col value;
data lines;
1 1 1 1
1 2 2 1
2 3 3 1
2 4 4 1
3 1 2 1
4 3 4 1
5 1 3 1
5 2 4 1
6 1 4 1
6 2 3 1
7 5 5 1
7 6 6 1;
run;

Figure S5. Syntax to set the parameter constraints outlined in Table S3.
Because gender is not a distinguishing variable in this example, it is no longer included in the MODEL statement as a fixed effect, nor does it interact with the stability or influence slopes. We use three new variables in this analysis: 1) partnum_1, which arbitrarily codes one dyad member as 1 and one dyad member as 2, 2) I1, which is coded 1 when partnum_1=1 and 0 when partnum_1=2, and 3) I2, which is coded 0 when partnum_1=1 and 1 when partnum_1=2.

```
PROC MIXED COVTEST; (1)
CLASS obs_id dyad partnum_1; (2)
MODEL pepreact_R = pep_lag_RC pep_lag_SC (3)
/CL S DDFM=satterth; (4)
RANDOM I1 I2 I1*pep_lag_RC I2*pep_lag_RC I1*pep_lag_SC I2*pep_lag_SC (5)
/GCORR SUB=dyad TYPE=lin(7) LDATA=hhh; (6)
REPEATED partnum_1 /TYPE=cs SUB=dyad*obs_id; (7)
PARMS (1.) (.4) (.0) (.0) (.0) (.0) (.0) (.2) (.2); (8)
run;
```

**Figure S6.** Syntax for a two-level crossed model with indistinguishable dyads.

In lines 2 and 7, “partnum_1” replaces “gender_class” as the variable that distinguishes between the dyad members of each dyad. In line 5, we list the variables to be used in the variance covariance matrix shown in Table S3. In line 6, TYPE=lin(7) specifies seven error covariances to be estimated for the variables listed and “LDATA=hhh” specifies that the matrix “hhh” will be used to place constraints on these parameters such that each covariance is the same for both members of the dyad. In line 8, The PARMS statement gives starting values for the covariance parameters. There must be a value for each covariance parameter in the model; in this case, there are nine: seven covariances, one common covariance, and one residual covariance. These can be adjusted to help the model converge.

Results for the fixed effects, which can be interpreted in the same manner as those in Example 1, are shown in Table S4. In this example, both the stability and influence paths are significant, indicating that receivers’ PEP reactivity is both a function of their own PEP reactivity at the prior time point and senders’ PEP reactivity at the prior time point. Influence in
this example is positive, meaning that higher values of sender PEP reactivity at one time point are associated with higher values of receiver PEP reactivity at the following time point.

**Table S4.** Fixed effects estimates.

<table>
<thead>
<tr>
<th>Effect</th>
<th>Estimate</th>
<th>SE</th>
<th>Df</th>
<th>T</th>
<th>p</th>
<th>Lower CI</th>
<th>Upper CI</th>
</tr>
</thead>
<tbody>
<tr>
<td>Intercept</td>
<td>-1.78</td>
<td>0.46</td>
<td>23.2</td>
<td>-3.87</td>
<td>&lt; .001</td>
<td>-2.74</td>
<td>-0.83</td>
</tr>
<tr>
<td>pep_lag_RC</td>
<td>0.39</td>
<td>0.03</td>
<td>26.2</td>
<td>13.39</td>
<td>&lt; .001</td>
<td>0.33</td>
<td>0.45</td>
</tr>
<tr>
<td>pep_lag_RC</td>
<td>0.05</td>
<td>0.02</td>
<td>41.4</td>
<td>2.05</td>
<td>.046</td>
<td>0.001</td>
<td>0.09</td>
</tr>
</tbody>
</table>

The estimates for the random effects are shown in Table S5. Note that the output from SAS will list the random effects as “LIN(1),” “LIN(2),” and so on. These numbers correspond to the matrix specified in Figure S5. These results allow us to answer questions such as whether there is variance in the extent to which participants are stable in PEP reactivity (yes), whether there is variance in the extent to which participants are influenced in their PEP reactivity (no), and whether PEP reactivities of two dyad members at the same time point are correlated (yes).
Table S5. Random effects estimates.

<table>
<thead>
<tr>
<th>Random effects ([co-]variances)</th>
<th>Estimate</th>
<th>SE</th>
<th>z</th>
<th>p</th>
</tr>
</thead>
<tbody>
<tr>
<td>Variance of intercept</td>
<td>12.99</td>
<td>3.00</td>
<td>4.32</td>
<td>&lt; .001</td>
</tr>
<tr>
<td>Variance of receiver lag</td>
<td>0.03</td>
<td>0.01</td>
<td>3.66</td>
<td>&lt; .001</td>
</tr>
<tr>
<td>Between-person covariance of one partner’s intercept with the other partner’s intercept</td>
<td>-1.79</td>
<td>2.88</td>
<td>-0.62</td>
<td>0.53</td>
</tr>
<tr>
<td>Between-person covariance of one partner’s receiver lag and the other partner’s sender lag</td>
<td>-0.01</td>
<td>0.01</td>
<td>-0.59</td>
<td>0.56</td>
</tr>
<tr>
<td>Within-person covariance of intercept with receiver lag</td>
<td>-0.17</td>
<td>0.12</td>
<td>-1.40</td>
<td>0.16</td>
</tr>
<tr>
<td>Between-person covariance of one partner’s intercept with the other partner’s receiver lag</td>
<td>-0.21</td>
<td>0.13</td>
<td>-1.65</td>
<td>0.10</td>
</tr>
<tr>
<td>Variance of sender lag</td>
<td>0.01</td>
<td>0.01</td>
<td>1.67</td>
<td>0.10</td>
</tr>
<tr>
<td>Common covariance</td>
<td>1.84</td>
<td>0.37</td>
<td>5.03</td>
<td>&lt; .001</td>
</tr>
<tr>
<td>Residual variance</td>
<td>10.24</td>
<td>0.43</td>
<td>23.90</td>
<td>&lt; .001</td>
</tr>
</tbody>
</table>

Graphing Influence

When physiological influence changes over time or is moderated by behaviors, a figure of these results may aid in presentation. To graph influence, only the influence coefficient and coefficients for interaction terms including influence should be used to create predicted influence values. For a basic example, we use the results from Table 10 to create the graph in Figure S7.

The Y axis represents predicted physiological influence values, which are slope values. Sender questions asked are represented on the X axis, at a low value (one standard deviation below the mean, which is -1.49) and at a high value (one standard deviation above the mean, which is 1.49). We only use the influence coefficients to create our four predicted values (influence at low sender questions asked and influence at high sender questions asked for both males and females): “pep_lag_SC,” “gender*pep_lag_SC,” “qasked_SC*pep_lag_SC,” and “gender*qasked_SC*pep_lag_SC.” Recall that gender is coded as -1 for females and 1 for males. We do not need to use the influence by receiver questions asked coefficients because the figure
we are creating is when receiver questions asked is at its mean, which equals zero. Therefore, for females, influence at low sender questions asked is calculated as \((.02)+(.04*-1)+(-.01*-1.49)+(-.01*-1*1.49)\) and influence at high sender questions asked is calculated as \((.02)+(.04*-1)+(-.01*1.49)+(-.01*1*1.49)\). For males, influence at low sender questions asked is calculated as \((.02)+(.04*1)+(-.01*-1.49)+(-.01*1*1.49)\) and influence at high sender questions asked is calculated as \((.02)+(.04*1)+(.01*1.49)+(-.01*1*1.49)\).

Figure S7. Physiological influence as a function of receiver gender and sender questions asked.

**Power Analysis**

We walk through a power analysis in SAS using data collected from 34 dyads composed of one man and one woman each across ten time points. This sample of 34 dyads is used as a pilot study to plan the sample size for the full study. The dependent variable in the following model is receiver PEP reactivity (‘pepreact_R’). Receivers’ own PEP reactivity at the prior time point (referred to as receiver lag) provides the stability path (“Zpep_lag_R”). Senders’ PEP reactivity at the prior time point (referred to as sender lag) provides the influence path (“Zpep_lag_S”).
The first step of the power analysis is to run the model one intends to use on the full sample of dyads on this pilot sample of 34 dyads. In Figure S8, we present this model. Note that for ease of presentation, we include only the stability and influence paths as fixed effects and estimate only a random stability path, but additional effects can be added to the model. “Dyad” is a unique identification number for each dyad, which is the same for each member of a dyad. “Obs_id” is a unique identification number for each pair of observations that occur at the same time point for the same dyad. It is calculated as “time + nt(dyad-1)” where “time” represents the time point of the observation, “nt” is the number of time points, and “dyad” is the unique identification number for each dyad. “Gender_class” is a variable representing gender, which is coded as -1 for females and 1 for males. “Male” is coded as 0 for females and 1 for males and “female” is coded as 0 for males and 1 for females. The receiver lag and sender lag variables have been standardized before conducting the analysis. The reason for this will become clear in the following step.

```
PROC MIXED COVTEST;
CLASS dyad obs_id gender_class;
MODEL pepreact_R = Zpep_lag_R Zpep_lag_S /S;
RANDOM female*Zpep_lag_R male*Zpep_lag_S /SUB=dyad TYPE=vc;
REPEATED gender_class /TYPE=csh SUB=dyad*obs_id;
RUN;
```

Figure S8. Syntax for Step 1 of the power analysis: conducting a two-level crossed model on a pilot sample of data.

Once the pilot data are analyzed, that information can be used to simulate data for 1000 hypothetical studies. In Figure S9, we show syntax for how to do this. In line 3, we chose an arbitrary number for the random number generator so that the simulation will output the same values every time. In lines 4 through 6, we specified the value for the intercept (line 4), the coefficient for receiver lag (line 5), and the coefficient for sender lag (line 6) that we obtained from the model in Figure S9. Note that the values in lines 4 through 6 are unstandardized. We
specified the value for the random receiver lag slope in line 7 for women and in line 8 for men. We specified the residual variance for women in line 9 and for men in line 10. The common covariance is listed in line 11.

In line 12, we indicated the number of samples to be generated, and in line 14, we indicated the number of dyads for which we wish to obtain power estimates. Initially, we suggest using the same number of dyads and time points in one’s pilot study to check the estimates, standard errors, and degrees of freedom obtained in the power analysis against the results obtained from the pilot study. This can help to identify mistakes in one’s syntax at this stage of the power analysis. In lines 15 and 16, we created the unique receiver lag slopes for each dyad—once for women and once for men. We indicated that there are two people per dyad in line 18, and in lines 19 through 24, we assigned those individuals to the appropriate value for the variables “gender_class,” “male,” and “female.”

In line 25, we specified that each person has ten time points of data (as in the pilot study). We chose numbers randomly from a standard normal distribution in lines 26 and 27 to be the receiver and sender lag variables. Because these numbers are selected from a standard normal distribution, we used standardized receiver and sender lag variables in the model conducted on the pilot data to ensure that our simulated data matches our pilot data as closely as possible. In lines 29 and 31, we generated a residual for women and men, and in lines 30 and 32, we specified the model, as used in Step 1 on the pilot data. Once this step has been completed, we recommend opening up the new file that has been generated (here it is called “example1”) and checking that the data appear as you anticipated—for example, that there are 34 dyads per sample, 2 people per dyad, and 10 time points per person, with unique X1 and X2 values at each time point.
Figure S9. Syntax for Step 2 of the power analysis: simulating data for 1000 hypothetical studies.

DATA example1; (1)
RETAIN SUBNUM; (2)
RETAIN SEED 3888; (3)
RETAIN B0 -4.5354; (4)
RETAIN B1 6.1527; (5)
RETAIN B2 -0.2944; (6)
RETAIN VSUBX1_F 1.4522; (7)
RETAIN VSUBX1_M 0.9626; (8)
RETAIN VRESID_F 16.6280; (9)
RETAIN VRESID_M 14.5752; (10)
RETAIN VCS 0.1990; (11)
RETAIN SAMPLES 1000; (12)
DO SAMPLE=1 TO SAMPLES; (13)
  DO DYAD=1 TO 34; (14)
    SUB_X1_F=SQRT (VSUBX1_F)*RANNOR(SEED); (15)
    SUB_X1_M=SQRT (VSUBX1_M)*RANNOR(SEED); (16)
    SUB_CS=SQRT (VCS)*RANNOR(SEED); (17)
    DO PERSON = 1 TO 2; (18)
      IF PERSON = 1 THEN GENDER_CLASS = -1; (19)
      IF PERSON = 2 THEN GENDER_CLASS = 1; (20)
      IF PERSON = 1 THEN MALE = 0; (21)
      IF PERSON = 2 THEN MALE = 1; (22)
      IF PERSON = 1 THEN FEMALE = 1; (23)
      IF PERSON = 2 THEN FEMALE = 0; (24)
    DO TIME = 0 TO 9; (25)
      X1 = RANNOR(SEED); (26)
      X2 = RANNOR(SEED); (27)
      obs_ID = TIME + (10*(DYAD-1)); (28)
      IF PERSON = 1 THEN RY_F = SQRT (VRESID_F)*RANNOR(SEED); (29)
      IF PERSON = 1 THEN Y = B0 + (B1+SUB_X1_F)*X1+B2*X2+SUB_CS+RY_F; (30)
      IF PERSON = 2 THEN RY_M = SQRT (VRESID_M)*RANNOR(SEED); (31)
      IF PERSON = 2 THEN Y = B0 + (B1+SUB_X1_M)*X1+B2*X2+SUB_CS+RY_M; (32)
    OUTPUT;
  END;
END;
END;
RUN;

Once the data have been simulated, Step 3 is to analyze each of those 1000 samples individually. We present syntax for this step in Figure S10. Note that the model directly replicates the model specified in the first step in Figure S8. Lines 9 and 10 of this syntax create two new data files—one with the results for the fixed effects for the analysis conducted on each sample and one with the results for the random effects for the analysis conducted on each sample.
The final step in this process is to go through the two new files created and count the number of times that an effect is significant (out of the 1000 samples). Syntax for this purpose is presented in Figure S11. Lines 1 through 6 provide syntax for looking at the fixed effects, and lines 7 through 16 provide syntax for looking at the random effects. For every two lines of syntax, SAS will output data that indicate the average estimate, standard error, degrees of freedom (for fixed effects), and the average amount of times an effect was significant. Figure S12 presents this output for X2 (produced by lines 5 and 6 of the syntax in Figure S11), which is the sender lag variable. The value in the “sig” line represents the power for the specified effect, given the parameters specified. In Figure S12, the power for detecting a significant effect of X2 is 48.10%, given the parameters specified.
data Intercept; set fixedeffects; if effect='Intercept'; if probt le .05 then sig=1; else sig=0; run; (1)
proc means data=Intercept; var estimate stderr df sig; title 'Intercept'; run; (2)
data X1; set fixedeffects; if effect='X1'; if probt le .05 then sig=1; else sig=0; run; (3)
proc means data=X1; var estimate stderr df sig; title 'X1'; run; (4)
data X2; set fixedeffects; if effect='X2'; if probt le .05 then sig=1; else sig=0; run; (5)
proc means data=X2; var estimate stderr df sig; title 'X2'; run; (6)
data X1xFEMALE; set randomeffects; if CovParm='X1xFEMALE'; if probt le .05 then sig=1; else sig=0; run; (7)
proc means data=X1xFEMALE; var estimate sig; title 'X1xFEMALE'; run; (8)
data X1xMALE; set randomeffects; if CovParm='X1xMALE'; if probt le .05 then sig=1; else sig=0; run; (9)
proc means data=X1xMALE; var estimate sig; title 'X1xMALE'; run; (10)
data Var1; set randomeffects; if CovParm='Var1'; if probt le .05 then sig=1; else sig=0; run; (11)
proc means data=Var1; var estimate sig; title 'Var1'; run; (12)
data Var2; set randomeffects; if CovParm='Var2'; if probt le .05 then sig=1; else sig=0; run; (13)
proc means data=Var2; var estimate sig; title 'Var2'; run; (14)
data CSH; set randomeffects; if CovParm='CSH'; if probt le .05 then sig=1; else sig=0; run; (15)
proc means data=CSH; var estimate sig; title 'CSH'; run; (16)

**Figure S11.** Syntax for Step 4 of the power analysis: counting the number of times each effect is significant.

<table>
<thead>
<tr>
<th>Variable</th>
<th>Label</th>
<th>N</th>
<th>Mean</th>
<th>Std Dev</th>
<th>Minimum</th>
<th>Maximum</th>
</tr>
</thead>
<tbody>
<tr>
<td>Estimate</td>
<td>Standard Error</td>
<td>1000</td>
<td>-0.2981438</td>
<td>0.1547205</td>
<td>-0.7639858</td>
<td>0.1446544</td>
</tr>
<tr>
<td>StdErr</td>
<td></td>
<td>1000</td>
<td>0.1550245</td>
<td>0.0059312</td>
<td>0.1360999</td>
<td>0.1754845</td>
</tr>
<tr>
<td>DF</td>
<td></td>
<td>1000</td>
<td>610.0000000</td>
<td>0</td>
<td>610.0000000</td>
<td>610.0000000</td>
</tr>
<tr>
<td>sig</td>
<td></td>
<td>1000</td>
<td>0.4810000</td>
<td>0.4998889</td>
<td>0</td>
<td>1.0000000</td>
</tr>
</tbody>
</table>

**Figure S12.** SAS output from lines 5 and 6 of Figure S11 in Step 4 of the power analysis.

As noted before, when first conducting a power analysis on a sample of data, we suggest checking the estimates, standard errors, and degrees of freedom obtained in the power analysis against the results obtained from the pilot study to help identify errors in one’s syntax. If the
results here match what was found on the sample of data, one can then move ahead and change parameters of the model in Step 2 (for example, number of dyads, time points, size of effects, etc.) to examine how these changes will impact the power obtained for various effects. If researchers have difficulty obtaining results that match what they found with their sample data, we recommend simplifying the model as much as possible, checking that results match, and then adding complexity from there. One reason results may not match is because of missing data. If there are missing data in the pilot study but missing data are not simulated in Step 2 (as we did not do here), researchers will find that the degrees of freedom will be different. Furthermore, if the variables in one’s pilot study are not normally distributed but values in the second step of the power analysis are chosen from a standard normal distribution, the results may also be slightly mismatched.